



Arrays Part IV

By Nate Moore, CPA, MBA, FACMPE

Hold on to your hat! This article adds a new degree of complexity to our discussion of arrays. Please review the first three articles in *RCM Advisor* to refresh your memory on arrays. Now that we have worked through several examples with the syntax of arrays and have seen how arrays can make spreadsheets easier, this article goes even further. The most important thing to remember as you follow the examples in this article is to enter the formulas with **Ctrl+Shift+Enter** to get the curly braces that surround Excel formulas. Manually typing the curly braces will not cause Excel to treat the formula as an array, so you will not get the results you expect. The more examples you see of the power of arrays, the more ways you can find to use arrays to make your spreadsheets more efficient.

Adding and Multiplying TRUE and FALSE

You may be familiar with Excel values of TRUE and FALSE from working with IF statements. The TRUE and FALSE values can also add significant power and flexibility once you understand how Excel does mathematical calculations with TRUE and FALSE. Figure 1 has a summary of the different ways to add and multiply TRUE and FALSE. You might find it helpful to think of TRUE being 1 and FALSE being 0. Just as $1 + 0 = 1$, $TRUE + FALSE = 1$. Similarly, TRUE times TRUE (Figure 1 uses the asterisk (*) as the times symbol) is 1, while TRUE times FALSE equals 0.

Now consider the table of data in Figure 2, which summarizes collections for the past four years for three locations of a medical practice. If the objective is to summarize 2014 collections, one formula that would work is `=SUMIF(A1:A15,E1,C1:C15)`, assuming cell E1 has 2014, the year we want to sum. Excel's SUMIF and SUMIFS formulas are described in Excel Videos 184 and 185 at mooresolutionsinc.com. The SUMIF and SUMIF functions were also discussed in the March/April 2018 issue of *RCM Advisor*. The array formula to do the same calculation may initially look more difficult, but arrays will give us more powerful filtering capability as our examples proceed.

To make the same calculation summing 2014 collections with an array formula, again assuming the year to sum is in

FIGURE 1

TRUE	+	FALSE	1
TRUE	*	FALSE	0
FALSE	+	FALSE	0
FALSE	*	FALSE	0
TRUE	+	TRUE	2
TRUE	*	TRUE	1

FIGURE 2

	A	B	C
1	2014	North	\$ 1,971,319
2	2014	South	\$ 1,617,303
3	2014	East	\$ 3,141,727
4	2015	North	\$ 3,266,638
5	2015	South	\$ 2,803,119
6	2015	East	\$ 1,264,411
7	2016	North	\$ 516,178
8	2016	South	\$ 536,613
9	2016	East	\$ 6,515,795
10	2017	North	\$ 523,728
11	2017	South	\$ 1,845,723
12	2017	East	\$ 4,874,361
13	2018	North	\$ 3,904,191
14	2018	South	\$ 1,873,783
15	2018	East	\$ 3,772,396

cell E1, the formula would be `{=SUM(IF(A1:A15=E1,C1:C15,0))}`. Working to understand the formula from the inside out, the IF formula checks to see if cells A1:A15 equal E1, the year in question. If the years match, the IF formula returns the corresponding collection amount in cells C1:C15. Otherwise, the IF formula returns 0. The SUM formula sums the results of the IF formula. Remember that a normal IF function only accepts one cell at a time. Because we use **Ctrl+Shift+Enter** to enter the formula, Excel treats the entire formula as an array and evaluates all 15 rows in the formula at once.

In a similar way, assume that the clinic location is stored in cell E2 and we want to sum all collections in the South location across years 2014 through 2018. A SUMIF formula would be `=SUMIF(B1:B15,E2,C1:C15)`. A similar array formula would be `{=SUM(IF(B1:B15=E2,C1:C15,0))}`. These formulas are very similar to summing years, and again, the SUMIF formula may look easier than the array formula.

FIGURE 3 All cells (Figures 1 and 2) at once.

	A	B	C	D	E	F	G	H	I	J	K		
1	2014	North	\$ 1,971,319		2014	\$ 6,730,349	\$ 6,730,349	SUMIF Excel Video 184					
2	2014	South	\$ 1,617,303		South	\$ 38,427,285	\$ 8,676,541						
3	2014	East	\$ 3,141,727			\$ 45,157,634							
4	2015	North	\$ 3,266,638				\$ 1,617,303	SUMIFS Excel Video 185					
5	2015	South	\$ 2,803,119	2014	AND	South	\$ 1,617,303						
6	2015	East	\$ 1,264,411	2014	OR	South	\$ 13,789,587						
7	2016	North	\$ 516,178										
8	2016	South	\$ 536,613			TRUE	+	FALSE	1				
9	2016	East	\$ 6,515,795			TRUE	*	FALSE	0				
10	2017	North	\$ 523,728			FALSE	+	FALSE	0				
11	2017	South	\$ 1,845,723			FALSE	*	FALSE	0				
12	2017	East	\$ 4,874,361			TRUE	+	TRUE	2				
13	2018	North	\$ 3,904,191			TRUE	*	TRUE	1				
14	2018	South	\$ 1,873,783										
15	2018	East	\$ 3,772,396	<code>{=SUM(IF((\$A\$1:\$A\$15=D6)+(\$B\$1:\$B\$15=F6)>0,\$C\$1:\$C\$15,0))}</code>									
16													

What if you want to sum all cells where the year is 2014 (cell D5) AND the location is South (cell F5)? The SUMIF formula now has multiple criteria (2014 and South), so we need to use a SUMIFS formula like =SUMIFS(\$C\$1:\$C\$15, \$A\$1:\$A\$15,D5,\$B\$1:\$B\$15,F5). A similar array formula for 2014 (cell E1) and South (cell E2) is {=SUM(IF((\$A\$1:\$A\$15=D5)*(\$B\$1:\$B\$15=F5)>0,\$C\$1:\$C\$15,0))}.

As scary as this formula may look, it is not very different from our first array formula. Again working from the inside out, the inner formula is an IF function. The IF function tests to see if this condition is true: (\$A\$1:\$A\$15=D5)*(\$B\$1:\$B\$15=F5)>0. The condition checks to see if cells A1:A15 are equal to D5, the year 2014. The condition also checks to see if cells B1:B15 are equal to F5, the South location. Remember that if the condition is TRUE, Excel treats the answer as 1. If the condition is FALSE, Excel treats the answer as 0. If the year in cells A1:A15 is 2014, the first half of the IF condition will be TRUE, or 1. If the location in cells B1:B15 is South, the second half of the IF condition will be TRUE, or 1. Since we are multiplying the results, only when both the year is 2014 and the location is South will the result be greater than 0, making the overall IF condition TRUE. When the IF condition is TRUE, the formula returns the corresponding cell in C1:C15. Otherwise, the formula returns 0. The outer SUM function sum the results of each calculation in the array.

Please note that for the SUMIFS formula to work, we need to look for both the year 2014 AND the location South. If we want to sum collections where the year is 2014 OR the location is South, the array's power and flexibility comes through for us. To sum rows where the year is 2014 OR the location is South, our array formula is almost identical to

the AND formula above: {=SUM(IF((\$A\$1:\$A\$15=D6)+(\$B\$1:\$B\$15=F6)>0,\$C\$1:\$C\$15,0))}. The only difference is that we add the two conditions instead of multiplying. The addition is highlighted in the OR array formula. Review Figure 1. When we wanted 2014 AND South, a TRUE value times a TRUE value will equal 1. If either value is FALSE, 1 * 0 or 0 * 1 will equal 0, and the row will be excluded from our AND calculation above. Now that we want 2014 OR South, we add. If either the year is 2014 or the location is South, the TRUE value will equal 1. By adding, we capture a 1 where either value is true and sum the corresponding row in column C in our formula.

This is clearly a complex array formula. You will not be the only one who re-reads a paragraph or two in this article. Once you understand the concept, you can build all kinds of complex AND or OR criteria to find exactly the conditions you need to in your data.

Continue to practice using array formulas with your data. Start small. Build test columns and take the array formulas a piece at a time to make sure you get the results you expect. Once the array formula plays nicely, arrays can add a lot of power and flexibility to your spreadsheets. There are more array examples in the Excel Video playlists at mooresolutionsinc.com ■

Nate Moore, CPA, MBA, FACMPE, writes custom SQL Server code to mine practice management data for analysis in Excel, web pages, and via email. Nate's second book, Better Data, Better Decisions – The SQL: Business Intelligence for Medical Practices, was recently published by MGMA. His free Excel Videos have been viewed over 2 million times and are available at mooresolutionsinc.com. Attend the HBMA Annual Conference to get hands-on training with Nate.